
Revit ファミリ API 実習 
 

    Lab1 – 角柱の作成 

March, 2010 by M. Harada 

Updated by DevTech AEC WG 

Last modified: 6/9/2017 

 

<VB.NET>VB.NETバージョン</VB.NET>  

目的:この実習では、ファミリ APIの基本を学習していきます。学習する項目は次のとおりです。 

         ファミリ環境をチェックする 

         押し出しを使用して単純なソリッドを作成する 

         位置合わせを設定する 

         タイプを加える 

タスク:長方形のプロファイルを持つ柱ファミリを作成するコマンドを定義して、異なる寸法のバリエーションを持つ

3つのタイプを加えます: 

(0)  テンプレートとして「柱(メートル単位).rft」 を使用するものとし、ユーザがこのテンプレートを開いていると

仮定する。コマンド内では、念のため、ユーザが正しいテンプレートを選んだかどうかチェックする  

(1)  長方形のプロファイルを定義して押し出し、単純なボックス ソリッドを作成する 

(2)  各面と対応する参照面の間に位置合わせを追加する 

(3)  寸法のバリエーションでタイプを定義する。   

図 1は、この実習に定義する角柱のイメージを示します。 

図 1.Lab1で作成する長方形のプロファイルを持った柱ファミリ 

この実習の実装と確認の手順は、下記のとおりです: 

1.    外部コマンドを定義する 

2.    ドキュメント コンテキストの有効性をチェックする 

3.    押し出しで単純なソリッドを作成する 

4.    位置合わせを追加する 

5.    タイプを追加する 

6.    作成した柱ファミリをテストする 

付録 A. Lab1 で使われるヘルパー関数  

1.   外部コマンドを定義する 

  

1.1  Visual Studio で開発言語(C#または VB.NET)を選択して新しいプロジェクトを作成して、参照を追加し

て外部コマンドを定義します。ファイル名とクラス名などは、ここでは VB.NET を選択して以下のように指定

します:   

         ソリューション名:  FamilyLabs 

         プロジェクト名称:  FamilyLabsVB 

         ファイル名:  1 FamilyCreateColumnRectangle.vb 

         コマンド クラス名:RvtCmd_FamilyCreateColumnRectangle  

  

(ここで希望する名前を使用しても構いません。ただし、その場合、プロジェクト名など、このドキュメント内で

は記述されている名称は、自分でつけた名称で代替して参照してください) 

  

次の参照が少なくとも必要となります:   

         System 

         System.Core(LINQ クエリ用) 

         System.Windows.Forms 

         Revit API 

         RevitAPIUI 

  

1.2  DBレベルのアプリケーションとドキュメントをそれぞれ保持する_rvtApp と_rvtDoc を定義します。下記はそ

の例です: 

<VB.NET> 

<Autodesk.Revit.Attributes.Transaction(Autodesk.Revit.Attributes.TransactionM

ode.Manual)> _ 

<Autodesk.Revit.Attributes.Regeneration(Autodesk.Revit.Attributes.Regeneratio

nOption.Manual)> _ 

Public Class RvtCmd_FamilyCreateColumnRectangle 

    Implements IExternalCommand 

    '' member variables for top level access to the Revit database 

    '' 

    Dim _rvtApp As Application 

    Dim _rvtDoc As Document 

  

    ''  command main 

    '' 

    Public Function Execute( _ 

        ByVal commandData As ExternalCommandData, _ 

        ByRef message As String, _ 

        ByVal elements As ElementSet) _ 

        As Result _ 

        Implements IExternalCommand.Execute 

  

        ''  objects for the top level access 

        '' 

        _rvtApp = commandData.Application.Application 

        _rvtDoc = commandData.Application.ActiveUIDocument.Document 

  

        ''  ... 

        '' 

  

        ''  finally, return 

        Return Result.Succeeded 

    End Function 

End Class 

</VB.NET> 

  

2. ドキュメント・コンテキストの有効性をチェックする 

ファミリ APIを利用するコマンド は、ファミリ エディタのコンテキスト内でのみで動作します。まず、現在開いている

ドキュメント コンテキストの有効性をチェックします。このチェックには、isRightTemplate()関数を定義して利用

します。isRightTemplate()関数は、引数として BuiltInCategoryオブジェクトを受け取ります。  

2.1 クラスに次の関数を追加します: 

<VB.NET> 

    Function isRightTemplate(ByVal targetCategory As BuiltInCategory) _ 

    As Boolean 

  

        ''  This command works in the context of family editor only. 

        '' 

        If Not _rvtDoc.IsFamilyDocument Then 

            MsgBox("This command works only in the family editor.") 

            Return False 

        End If 

  

        ''  Check the template for an appropriate category here if needed. 

        '' 

        Dim cat As Category = 

_rvtDoc.Settings.Categories.Item(targetCategory) 

        If _rvtDoc.OwnerFamily Is Nothing Then 

            MsgBox("This command only works in the family context.") 

            Return False 

        End If 

        If Not cat.Id.Equals(_rvtDoc.OwnerFamily.FamilyCategory.Id) Then 

            MsgBox("Category of this family document does not match the 

context required by this command.") 

            Return False 

        End If 

  

        ''  if we come here, we should have a right one. 

        Return True 

    End Function 

</VB.NET> 

  

この関数は、次をチェックします:   

         ファミリ ドキュメントかどうか。_rvtDoc.IsFamilyDocumentでこれをチェックできます。 

         テンプレートが柱ファミリ定義用かどうか。  現在のドキュメントのカテゴリは

_rvtDoc.OwnerFamily.FamilyCategory.Id を見て確認することができます。 

上記の条件が満たされる場合、関数は true を返し、そうでなければ false を返すものとします。   

2.2 メイン コマンドの Execute() 関数から、isRightTemplate() 関数を呼び出します。ドキュメントが正しいテンプ

レートでない場合、コマンドは実行を停止します: 

<VB.NET> 

        _rvtApp = commandData.Application.Application 

        _rvtDoc = commandData.Application.ActiveUIDocument.Document 

  

        ''  (0) This command works in the context of family editor only. 

        ''      We also check if the template is for an appropriate category 

if needed. 

        ''      Here we use a Column(i.e., Metric Column.rft) template. 

        ''      Although there is no specific checking about metric or 

imperial, our lab only works in metric for now. 

        '' 

        If Not isRightTemplate(BuiltInCategory.OST_Columns) Then 

            MsgBox("Please open 柱(メートル単位).rft") 

            Return IExternalCommand.Result.Failed 

        End If 

  

        ... 

</VB.NET> 

 

 

 



 6 / 17 

 

押し出しで単純なソリッドを作成する 

テンプレートが有効だと判断できたら、単純なソリッドの作成を開始します。ここで、長方形のプロファイルを定義

して、与えられた高さでそれを押し出します。  

 3.1 クラスに次の関数を加えてください。 このコードは、単純な矩形形状を持つプロファイルを定義します: 

<VB.NET> 

    '' ============================================ 

    ''   (1.1) create a simple rectangular profile 

    '' ============================================ 

    Function createProfileRectangle() As CurveArrArray 

  

        '' 

        ''  define a simple rectangular profile 

        '' 

        ''  3     2 

        ''   +---+ 

        ''   |   | d    h = height 

        ''   +---+ 

        ''  0     1 

        ''  4  w 

        '' 

  

        ''  sizes (hard coded for simplicity) 

        ''  note: these need to match reference plane. otherwise, alignment 

won't work. 

        ''  as an exercise, try changing those values and see how it behaves. 

        '' 

        Dim w As Double = mmToFeet(600) ' hard coded for simplicity here. in 

practice, you may want to find out from the references) 

        Dim d As Double = mmToFeet(600) 

  

        ''  define vertices 

        '' 

        Const nVerts As Integer = 4 '' the number of vertices 

        Dim pts() As XYZ = {New XYZ(-w / 2, -d / 2, 0), New XYZ(w / 2, -d / 2, 

0), New XYZ(w / 2, d / 2, 0), New XYZ(-w / 2, d / 2, 0), New XYZ(-w / 2, -d / 

2, 0)} ' the last one is to make the loop simple 

  

        ''  define a loop. define individual edges and put them in a 

curveArray 

        '' 

        Dim pLoop As CurveArray = _rvtApp.Create.NewCurveArray 

        Dim lines(nVerts - 1) As Line 

        For i As Integer = 0 To nVerts - 1 

            lines(i) = Line.CreateBound(pts(i), pts(i + 1)) 

            pLoop.Append(lines(i)) 

        Next 

  

        ''  then, put the loop in the curveArrArray as a profile 

        '' 

        Dim pProfile As CurveArrArray = _rvtApp.Create.NewCurveArrArray 

        pProfile.Append(pLoop) 

        ''  if we come here, we have a profile now. 

  

        Return pProfile 

  

    End Function 

</VB.NET> 

  

この関数内で使用するヘルパー関数が一つがあります。 

 mmToFeet() 

Revitはアプリケーション内部では単位にフィートを使用します。APIによって寸法を定義する場合は、常に、単

位を変換する必要があります。 mmToFeet() のコードは、このドキュメントの終りにあるセクション 付録 A に記

述されています。このクラスの終わりに コピー&ペースト してください。 

この実習では、API の学習を目的とし、コードの判読性を優先するために、長方形の頂点と実際の寸法をハー

ドコーディングしています。サイズは、実際には柱ファミリ テンプレートの中で事前に定義される参照面間の距離

から取得します。異なるテンプレートを使用している場合は、それらの値を調節する必要があるでしょう。  プロフ

ァイルは CurveArrArray(あるいはカーブ配列のコレクション)として定義されます。   

テンプレートを開いて、下記の項目を確認してみましょう:  

         テンプレート内で使用される寸法 

         テンプレートに事前に用意された参照面の名前 

         平面図に正面図での見た目(表現)  

実習の全体を通して、それらを参照することに注意してください。 

3.2 上記で定義したプロファイルを使用して、押し出しからソリッドを作成します。下記の関数をクラス コードに加

えてください: 

</VB.NET> 

    '' ============================================ 
    ''   (1) create a simple solid by extrusion 
    '' ============================================ 
    Function createSolid() As Extrusion 
  
        '' 
        ''  (1) define a simple rectangular profile 
        '' 
        ''  3     2 
        ''   +---+ 
        ''   |   | d    h = height 
        ''   +---+ 

        ''  0     1 
        ''  4  w 
        '' 
        Dim pProfile As CurveArrArray = createProfileRectangle() 
        '' 
        ''  (2) create a sketch plane 
        '' 
        ''  we need to know the template. If you look at the template (Metric 

Column.rft) and "Front" view, 
        ''  you will see "Reference Plane" at "Lower Ref. Level". We are 

going to create an extrusion there. 
        ''  findElement() is a helper function that find an element of the 

given type and name.  see below. 
        '' 
        Dim pRefPlane As ReferencePlane = findElement(GetType(ReferencePlane), 

"参照面") ' need to know from the template 
        Dim pSketchPlane As SketchPlane = SketchPlane.Create(_rvtDoc, 

pRefPlane.Plane) 
  
        ''  (3) height of the extrusion 
        '' 
        ''  once again, you will need to know your template. unlike UI, the 

alightment will not adjust the geometry. 
        ''  You will need to have the exact location in order to set 

alignment. 
        ''  Here we hard code for simplicity. 4000 is the distance between 

Lower and Upper Ref. Level. 
        ''  as an exercise, try changing those values and see how it behaves. 
        '' 
        Dim dHeight As Double = mmToFeet(4000) '' distance between Lower and 

Upper Ref Level. 
  
        ''  (4) create an extrusion here. at this point. just an box, nothing 

else. 
        '' 
        Dim bIsSolid As Boolean = True ' as oppose to void. 
        Dim pSolid As Extrusion = _rvtDoc.FamilyCreate.NewExtrusion(bIsSolid, 

pProfile, pSketchPlane, dHeight) 
  
        Return pSolid 
  
    End Function 

</VB.NET> 
  

この関数の中では、新たに 2つめのヘルパー関数が使用されています。 

findElement() 

このヘルパー関数は、引数に与えられたタイプと名前から要素を検索します。  例えば、参照面、レベルやビュー

を見つけるために使用することが出来ます。 完全なコードは、このドキュメントの終わりに記述されています(付録 

A)。このクラスの終わりに コピー&ペースト してください。 

上記コードの一番下で、以下のメソッドを呼び出しています。 

_rvtDoc.FamilyCreate.NewExtrusion()。 

これは、押し出しを定義するために使用する主要なメソッドです。引数として、Solid/Void フラグ、プロファイル、

スケッチ平面と高さを使用します。  スケッチ平面を定義するためには、定義済みの参照面のうちの 1つを使用

します(例、上記コードやテンプレート内の”参照面”)。 

ここでは、再度、高さ情報をハードコーディングしています。この値はテンプレート由来で、上部と下部の参照レ

ベル間の距離です(例えば、テンプレートの正面図の中でこれを確認してみてください)。  

3.3 メインのコマンド機能から上記の関数を呼び出してください。 createSolid () 関数は、押し出しタイプのオブ

ジェクトを返します: 

<VB.NET> 

    Public Function Execute(ByVal commandData As ExternalCommandData,  
        ...  
  
        '' (0) This command works in the context of family editor only.  
        ... 
         
        If Not isRightTemplate(BuiltInCategory.OST_Columns) Then 
            MsgBox("Please open Metric Column.rft") 
            Return IExternalCommand.Result.Failed 
        End If 
  
        '' (1) create a simple extrusion. just a simple box for now.  
        Dim pSolid As Extrusion = createSolid() 
  
        '' We need to regenerate so that we can build on this new geometry 
        _rvtDoc.Regenerate() 
  
  
        ... 

</VB.NET> 
  

3.4.ソリッドが正確に作成されるかどうかテストするため、この時点でコードをビルドし、実行する準備が整いまし

た。どのようにソリッドが作成されるか試してみてください。  

  

下記のようにアドイン マニフェスト ファイルを作成して、それを Revit のアドインディレクトリに配置します。もちろん、

お使いの環境にそれぞれのパラメータを調節する必要があります。ここで、VisibilityMode が

「 NotVisibleInProject 」に設定されている点に注意してください。 これは、実装したコマンドが、Revit プロジェ

クトとしてではなく、ファミリ エディタ モードで動作することを意味します。   

 
 
<?xml version="1.0" encoding="utf-8" standalone="no"?> 

<RevitAddIns> 

  <AddIn Type="Command"> 

    <Assembly>FamilyVb.dll</Assembly> 

    <AddInId>98F162FC-90E3-411e-BEC8-75D403BBF11A</AddInId> 

    <FullClassName>FamilyVb.RvtCmd_FamilyCreateColumnRectangle</FullClassName

> 

    <Text>Family Labs Create Rectangular Column</Text> 

    <Description>Family API lab 1 to create rectangular column</Description> 

    <VisibilityMode>NotVisibleInProject</VisibilityMode> 

    <AccessibilityClassName>Revit.Samples.SampleAccessibilityCheck </Accessib

ilityClassName> 

    <VendorId>ADNP</VendorId> 

    <VendorDescription>Autodesk, Inc. www.autodesk.com</VendorDescription> 

  </AddIn> 

</RevitAddIns> 

 
  

「柱(メートル単位).rft」テンプレートを使用して、ファミリ エディタを起動してください。   

  

コマンドを実行した後で、柱が意図したように動作しないかもしれません。  これは、この段階では予想通りの振

る舞いです。 ファミリはパラメトリック のオブジェクトです。 上記で定義したソリッドは、モデルの初期状態です。 こ

れから、この柱ファミリにパラメトリックな振る舞いをさせるように機能を追加していきます。 

  
  

4. 位置合わせの追加 

次のステップは、ソリッドの各面に対応する参照面との間に位置合わせ拘束を追加します。これは、柱をパラメト

リック方式で動作させるために必要です。これにより、ユーザが柱の寸法を変更すると、サイズが自動調節され

る機能が追加されます。 

4.1 次の関数をクラスに加えてください。この関数は、6つの位置合わせの処理を追加します(ボックス形状の6つ

の各面に): 

  
<VB.NET> 

    '' ============================================ 
    ''   (2) add alignments 
    '' ============================================ 
    Sub addAlignments(ByVal pBox As Extrusion) 
  
        '' 
        ''  (1) we want to constrain the upper face of the column to the 

"Upper Ref Level" 
        '' 
  
        ''  which direction are we looking at? 
        '' 

        Dim pView As View = findElement(GetType(View), "正面") 
  
        ''  find the upper ref level 
        ''  findElement() is a helper function. see below. 
        '' 

        Dim upperLevel As Level = findElement(GetType(Level), "上参照レベル") 
        Dim ref1 As Reference = upperLevel.GetPlaneReference() 

        ''  find the face of the box 
        ''  findFace() is a helper function. see below. 
        '' 
        Dim upperFace As PlanarFace = findFace(pBox, New XYZ(0, 0, 1)) ' find 

a face whose normal is z-up. 
        Dim ref2 As Reference = upperFace.Reference 
  
        ''  create alignments 
        '' 
        _rvtDoc.FamilyCreate.NewAlignment(pView, ref1, ref2) 
  
        '' 
        ''  (2) do the same for the lower level 
        '' 
  
        ''  find the lower ref level 
        ''  findElement() is a helper function. see below. 
        '' 

        Dim lowerLevel As Level = findElement(GetType(Level), "下参照レベル") 
        Dim ref3 As Reference = lowerLevel.GetPlaneReference() 
  
        ''  find the face of the box 
        ''  findFace() is a helper function. see below. 
        '' 
        Dim lowerFace As PlanarFace = findFace(pBox, New XYZ(0, 0, -1)) ' 

find a face whose normal is z-down. 
        Dim ref4 As Reference = lowerFace.Reference 
  
        '' create alignments 
        '' 
        _rvtDoc.FamilyCreate.NewAlignment(pView, ref3, ref4) 
  
        '' 
        ''  (3)  same idea for the Right/Left/Front/Back 
        '' 
        ''  get the plan view 
        ''  note: same name maybe used for different view types. either one 

should work. 

        Dim pViewPlan As View = findElement(GetType(ViewPlan), "下参照レベル") 
  
        ''  find reference planes 
        Dim refRight As ReferencePlane = findElement(GetType(ReferencePlane), 

"右") 
        Dim refLeft As ReferencePlane = findElement(GetType(ReferencePlane), 

"左") 
        Dim refFront As ReferencePlane = findElement(GetType(ReferencePlane), 

"正面") 
        Dim refBack As ReferencePlane = findElement(GetType(ReferencePlane), 

"背面") 
  
        ''  find the face of the box 
        Dim faceRight As PlanarFace = findFace(pBox, New XYZ(1, 0, 0)) 

        Dim faceLeft As PlanarFace = findFace(pBox, New XYZ(-1, 0, 0)) 
        Dim faceFront As PlanarFace = findFace(pBox, New XYZ(0, -1, 0)) 
        Dim faceBack As PlanarFace = findFace(pBox, New XYZ(0, 1, 0)) 
  
        ''  create alignments 
        '' 
        _rvtDoc.FamilyCreate.NewAlignment(pViewPlan, refRight.GetReference(), 

faceRight.Reference) 
        _rvtDoc.FamilyCreate.NewAlignment(pViewPlan, refLeft.GetReference(), 

faceLeft.Reference) 
        _rvtDoc.FamilyCreate.NewAlignment(pViewPlan, refFront.GetReference(), 

faceFront.Reference) 
        _rvtDoc.FamilyCreate.NewAlignment(pViewPlan, refBack.GetReference(), 

faceBack.Reference) 
  
    End Sub 

</VB.NET> 

  

この関数内では、新たに 3つめのヘルパー関数が使用されています。 

findFace() 

このヘルパー関数は、押し出しソリッドから与えられた法線で平面を見つけます。完全なコードは、このドキュメン

トの終わりに記載されています(付録 A)。このクラスの終わりに コピー&ペースト してください。 

ソリッドの上部の面と参照面「上参照レベル」の間に位置合わせを追加するところで、コードの最初の部分に注

目してください。ここで1つ理解できれば、コードの残り部分でも同じように理解できるはずです。  

  
_rvtDoc.FamilyCreate.NewAlignment(pView、ref1、ref2) 

  

これは、新しい位置合わせを作成する主要なメソッドです。引数としてビューと2つの参照を渡します。UIで行う

ように、上部の面を上参照面への位置合わせするために、側面からモデルを見ています。ここでは、正面ビューを

使用します。  findElement () と findFace( ) はヘルパー関数です。   

  

注意すべきなのは、UI と異なり、API メソッドである NewAlignment() は、自動的に計算してモデルのジオメト

リを調節しないという点です。下記は、RevitAPI.chm ファイルからの抜粋です:   

  
“These references must be already geometrically aligned (this function will not force 
them to become aligned).” 

  

NewAlignment() を呼に出す前に、それらが同じ位置にあることを確かめる必要があります。   

  

4.2 メイン コマンドの Execute() から、addAlignments(pSolid) を呼び出します: 

  
<VB.NET> 

    Public Function Execute(ByVal commandData As ExternalCommandData, ByRef 
  
        ... 
  
        '' (1) create a simple extrusion. just a simple box for now.  



 13 / 17 

 

        Dim pSolid As Extrusion = createSolid() 
  
        '' (2) add alignment  
        addAlignments(pSolid) 
  
        ... 
</VB.NET> 

  

この時点でコードをビルドし、実行する準備が整いました。この柱は、参照を変更した際に正しく応答するはず

です(例えば、レベルなど)。 

  

5. タイプを追加する 

「幅」と「奥行」の寸法を変えたタイプを 2つ追加します： 

         600 x 900 

         1000×300 

         600 x 600 

5.1次の関数をクラスに加えてください: 

  
<VB.NET> 

    '' ============================================ 
    ''   (3) add types 
    '' ============================================ 
    Sub addTypes() 
  
        ''  addType(name, Width, Depth) 
        '' 
        addType("600x900", 600.0, 900.0) 
        addType("1000x300", 1000.0, 300.0) 
        addType("600x600", 600.0, 600.0) 
  
    End Sub 
  
    ''  add one type 
    '' 
    Sub addType(ByVal name As String, ByVal w As Double, ByVal d As Double) 
  
        ''  get the family manager from the current doc 
        Dim pFamilyMgr As FamilyManager = _rvtDoc.FamilyManager 
  
        ''  add new types with the given name 
        '' 
        Dim type1 As FamilyType = pFamilyMgr.NewType(name) 
  
        ''  look for 'Width' and 'Depth' parameters and set them to the given 

value 
        '' 

        ''  first 'Width' 
        '' 

        Dim paramW As FamilyParameter = pFamilyMgr.Parameter("幅") 
        Dim valW As Double = mmToFeet(w) 
        If paramW IsNot Nothing Then 
            pFamilyMgr.Set(paramW, valW) 
        End If 
  
        ''  same idea for 'Depth' 
        '' 

        Dim paramD As FamilyParameter = pFamilyMgr.Parameter("奥行") 
        Dim valD As Double = mmToFeet(d) 
        If paramD IsNot Nothing Then 
            pFamilyMgr.Set(paramD, valD) 
        End If 
  
    End Sub 

</VB.NET> 

  

ここで覚えておく必要のあるクラスは、ドキュメントのファミリ マネージャ オブジェクトです。このオブジェクトは次のよ

うにアクセスすることができます:  

  
_rvtDoc.FamilyManager 

  

一度、ファミリマネージャを取得したら、NewType メソッドを使用して、新しいタイプを作成することができます: 

  
pFamilyMgr.NewType(name) 

  

次のメンバ関数を使用して、当該パラメータにアクセスすることができます: 

  

             pFamilyMgr.Parameter("幅") 

  

その後で、次のように値を設定します: 

  
pFamilyMgr.Set(paramW、valW) 

  

上記のコードは3つのタイプを定義します。   

  

5.2 メイン コマンドの Execute() から、addTypes() を呼び出します: 

  
<VB.NET> 

    Public Function Execute(ByVal commandData As ExternalCommandData, ByRef  
  
        ... 
        '' (2) add alignment  
        addAlignments(pSolid) 
  
        '' (3) add types  
        addTypes() 

        ''  finally, return  
        Return Result.Succeeded 
  
    End Function 

</VB.NET> 

 
 

5.3 コードをビルドし、実行してみましょう。   

  
 

6. 作成した柱ファミリをテストする  

柱ファミリが正しく作成されていれば、この時点でコードをビルドし、実行することができます。 

  

「柱(メートル単位).rft」テンプレートを使用して、ファミリ エディタを起動してください。  

  

コマンドを実行した後に、タイプ ダイアログで3つのタイプが作成されるかどうかチェックして、それを適用して柱がサ

イズに従って変更されるか確かめてください。   

  

次の実習では、柱のプロファイルを修正して、参照面、パラメータと寸法を追加する方法を学習します。  

  
  

付録 A. Lab1の中で使われるヘルパー関数 

Lab1では、次のヘルパー関数を使用します。 必要に応じて下記のコードをコピー&ペーストで作成中のコード

に貼り付けてください。 

         findFace() － -与えられた押し出しソリッドから、与えられた法線で平面を見つける 

         findElement() － -与えられたタイプおよび名前の要素を見つける(例、参照やレベルの検索など)  

         mmToFeet() －  単位をミリメートルからフィートへ変換する 

<VB.NET> 

    ''============================================ 
    '' 
    ''  Helper functions 
    '' 
    ''============================================ 
  
#Region "Helper Functions" 
  
    '' ============================================ 
    ''   helper function: find a planar face with the given normal 
    '' ============================================ 
    Function findFace(ByVal pBox As Extrusion, ByVal normal As XYZ) As 

PlanarFace 
  
        '' get the geometry object of the given element 

        '' 
        Dim op As New Options 
        op.ComputeReferences = True 
        Dim geomObjs As IEnumerable(Of GeometryObject) = 

pBox.Geometry(op).AsEnumerable() 
        '' loop through the array and find a face with the given normal 
        '' 
        For Each geomObj As GeometryObject In geomObjs 
  
            If TypeOf geomObj Is Solid Then  ''  solid is what we are 

interested in. 
  
                Dim pSolid As Solid = geomObj 
                Dim faces As FaceArray = pSolid.Faces 
  
                For Each pFace As Face In faces 
                    Dim pPlanarFace As PlanarFace = pFace 
                    If Not (pPlanarFace Is Nothing) Then 
                        If pPlanarFace.FaceNormal.IsAlmostEqualTo(normal) 

Then '' we found the face 
                            Return (pPlanarFace) 
                        End If 
                    End If 
                Next 
  
            ElseIf TypeOf geomObj Is GeometryInstance Then 
  
                '' will come back later as needed. 
  
            ElseIf TypeOf geomObj Is Curve Then 
  
                '' will come nack later as needed. 
  
            ElseIf TypeOf geomObj Is Mesh Then 
  
                '' will come back later as needed. 
  
            Else 
                '' what else do we have? 
  
            End If 
        Next 
  
        '' if we come here, we did not find any. 
        Return Nothing 
  
    End Function 
  
    '' ============================================ 
    ''   helper function: find an element of the given type and the name. 
    ''   You can use this, for example, to find Reference or Level with the 

given name. 
    '' ============================================ 
    Function findElement(ByVal targetType As Type, ByVal targetName As 

String) As Autodesk.Revit.DB.Element 
  
        '' get the elements of the given type 
        '' 
        Dim collector = New FilteredElementCollector(_rvtDoc) 
        collector.WherePasses(New ElementClassFilter(targetType)) 
  
        '' parse the collection for the given name 
        '' using LINQ query here.    
        ''  
        Dim targetElems = From element In collector Where 

element.Name.Equals(targetName) Select element 
        Dim elems As List(Of Element) = targetElems.ToList() 
  
        If elems.Count > 0 Then '' we should have only one with the given 

name.   
            Return elems(0) 
        End If 
  
        '' cannot find it. 
        Return Nothing 
  
    End Function 
  
    '' ============================================ 
    ''   convert millimeter to feet 
    '' ============================================ 
    Function mmToFeet(ByVal mmVal As Double) As Double 
  
        Return mmVal / 304.8 '' * 0.00328; 
  
    End Function 
#End Region 

</VB.NET> 

